

# Introduction aux Systèmes d'Exploitation

## **Unit 7: Loops, Conditions, and Addressing Modes**

François Taïani



# Outline

- Loops
- If statements
- Addressing Modes
- Bonus

# Typical Construct : Loop

- Reminder : `cmp, jle, mov, dec`
- How would you implement the following in x86 asm?  
→ (using the `rbx` register for `i`)

```
int i;
```

```
for(i=15; i>0;i--) {  
    // Do something  
    printf("%i\n",i);  
} // EndFor
```



# Typical Construct : Loop

```
mov      rbx, 0xf          ; start value: i = 15
begin:   cmp      rbx, 0    ; if i <= 0 then go to end
jle      end
```

```
dec      rbx
jmp      begin
end:
...
```

# Typical Construct : Loop

```
SECTION    .data
message:   db      " ", 10      ; note the newline at the end
msgLen:    equ $-message

SECTION    .text
GLOBAL    _start
_start:
    mov     rbx, 0xf             ; start value: i = 15
begin:
    cmp     rbx, 0               ; if i <= 0 then go to end
    jle     end
    mov     rcx, rbx             ; we use rcx to compute the ascii code
    add     rcx, '0'             ; of the character we want to print
    mov     [message], cl        ; and move this character into the message string
    mov     rax, 1               ; system call for write
    mov     rdi, 1               ; file handle 1 is stdout
    mov     rsi, message         ; address of string to output
    mov     rdx, msgLen          ; number of bytes
    syscall                     ; invoke operating system to do the write
    dec     rbx                  ; no need to save/restore rbx: done in sys. call.
    jmp     begin
end:
...
```

# Making sense of the output

|   | 0                   | 1                   | 2                   | 3                   | 4                   | 5                   | 6                   | 7                   | 8                   | 9                  | A                   | B                   | C                  | D                  | E                  | F                  |   |
|---|---------------------|---------------------|---------------------|---------------------|---------------------|---------------------|---------------------|---------------------|---------------------|--------------------|---------------------|---------------------|--------------------|--------------------|--------------------|--------------------|---|
| 0 | 00 0000<br>NUL<br>□ | 01 0001<br>SOH<br>┐ | 02 0000<br>STX<br>└ | 03 0000<br>ETX<br>┘ | 04 0000<br>EOT<br>↘ | 05 0000<br>ENQ<br>☒ | 06 0000<br>ACK<br>✓ | 07 0000<br>BEL<br>⌚ | 08 0000<br>BS<br>↩  | 09 0000<br>HT<br>➤ | 10 0000<br>LF<br>≡  | 11 0000<br>VT<br>▼  | 12 0000<br>FF<br>⇓ | 13 0000<br>CR<br>⇐ | 14 0000<br>SO<br>⊗ | 15 0000<br>SI<br>⊙ | 8 |
| 1 | 16 0001<br>DLE<br>⊞ | 17 0001<br>DC1<br>⌚ | 18 0001<br>DC2<br>⌚ | 19 0001<br>DC3<br>⌚ | 20 0001<br>DC4<br>⌚ | 21 0001<br>NAK<br>↗ | 22 0001<br>SYN<br>⌒ | 23 0001<br>ETB<br>┘ | 24 0001<br>CAN<br>⊗ | 25 0001<br>EM<br>⋮ | 26 0001<br>SUB<br>? | 27 0001<br>ESC<br>⊖ | 28 0001<br>FS<br>⊞ | 29 0001<br>GS<br>⊞ | 30 0001<br>RS<br>⊞ | 31 0001<br>US<br>⊞ | 9 |
| 2 | 32 0010<br>SP       | 33 0010<br>!        | 34 0010<br>"        | 35 0010<br>#        | 36 0010<br>\$       | 37 0010<br>%        | 38 0010<br>&        | 39 0010<br>'        | 40 0010<br>(        | 41 0010<br>)       | 42 0010<br>✱        | 43 0010<br>+        | 44 0010<br>,       | 45 0010<br>-       | 46 0010<br>.       | 47 0010<br>/       | A |
| 3 | 48 0011<br>0        | 49 0011<br>1        | 50 0011<br>2        | 51 0011<br>3        | 52 0011<br>4        | 53 0011<br>5        | 54 0011<br>6        | 55 0011<br>7        | 56 0011<br>8        | 57 0011<br>9       | 58 0011<br>:        | 59 0011<br>;        | 60 0011<br><       | 61 0011<br>=       | 62 0011<br>>       | 63 0011<br>?       | B |
| 4 | 64 0100<br>@        | 65 0100<br>A        | 66 0100<br>B        | 67 0100<br>C        | 68 0100<br>D        | 69 0100<br>E        | 70 0100<br>F        | 71 0100<br>G        | 72 0100<br>H        | 73 0100<br>I       | 74 0100<br>J        | 75 0100<br>K        | 76 0100<br>L       | 77 0100<br>M       | 78 0100<br>N       | 79 0100<br>O       | C |
| 5 | 80 0101<br>P        | 81 0101<br>Q        | 82 0101<br>R        | 83 0101<br>S        | 84 0101<br>T        | 85 0101<br>U        | 86 0101<br>V        | 87 0101<br>W        | 88 0101<br>X        | 89 0101<br>Y       | 90 0101<br>Z        | 91 0101<br>[        | 92 0101<br>\<br>]  | 93 0101<br>^       | 94 0101<br>_       | 95 0101<br>`       | D |
| 6 | 96 0110<br>`        | 97 0110<br>a        | 98 0110<br>b        | 99 0110<br>c        | 100 0110<br>d       | 101 0110<br>e       | 102 0110<br>f       | 103 0110<br>g       | 104 0110<br>h       | 105 0110<br>i      | 106 0110<br>j       | 107 0110<br>k       | 108 0110<br>l      | 109 0110<br>m      | 110 0110<br>n      | 111 0110<br>o      | E |
| 7 | 112 0111<br>p       | 113 0111<br>q       | 114 0111<br>r       | 115 0111<br>s       | 116 0111<br>t       | 117 0111<br>u       | 118 0111<br>v       | 119 0111<br>w       | 120 0111<br>x       | 121 0111<br>y      | 122 0111<br>z       | 123 0111<br>{       | 124 0111<br>       | 125 0111<br>}      | 126 0111<br>~      | 127 0111<br>DEL    | F |

# Optimization (Special Case)

```
mov     rbx, 0xf           ; start value: 15
begin:

dec     rbx
jnz     begin              ; if rbx!=0 then loop back to begin
end:
...
```

# Optimization (Special Case)

```
SECTION    .data
message:   db      " ", 10      ; note the newline at the end
msgLen:    equ $-message

SECTION    .text
GLOBAL    _start
_start:
    mov     rbx, 0xf             ; start value: 15
begin:
    mov     rcx, rbx             ; we use rcx to compute the ascii code
    add     rcx, '0'             ; of the character we want to print
    mov     [message], cl        ; and move this character into the message string


    mov     rax, 1               ; system call for write
    mov     rdi, 1               ; file handle 1 is stdout
    mov     rsi, message         ; address of string to output
    mov     rdx, msgLen          ; number of bytes
    syscall                     ; invoke operating system to do the write
    dec     rbx
    jnz     begin                ; if rbx!=0 then loop back to begin
end:
    ...
```



# Can you spot the difference?

```
SECTION      .data
message:     db      " ", 10
msgLen:      equ  $-message
```


; note the newline at the end



```
SECTION      .text
GLOBAL      _start
_start:
```

```
    mov      rcx, 0xf          ; start value: 15
begin:
    mov      rbx, rcx          ; we use rcx to compute the ascii code
    add      rbx, '0'          ; of the character we want to print
    mov      [message], bl     ; and move this character into the message string
    push     rcx               ; (!) saving rcx, as clobbered by 'syscall'
    mov      rax, 1            ; system call for write
    mov      rdi, 1            ; file handle 1 is stdout
    mov      rsi, message      ; address of string to output
    mov      rdx, msgLen       ; number of bytes
    syscall                   ; invoke operating system to do the write
    pop      rcx               ; (!) restoring rcx
    loop     begin             ; equivalent to 'dec rcx' + 'jnz begin'
end:
    ...
```

# Can you spot the difference?



```
mov      rcx, 0xf  
begin:
```

```
; start value: 15
```

```
loop     begin  
end:  
...
```

```
; equivalent to 'dec rcx' + 'jnz begin'
```

# Typical Construct: If

- How would you implement the following in x86?

```
int x = 5 ;  
int y = 10;  
  
if (x < y) {  
    printf("X smaller than Y");  
} else {  
    printf("X greater or equal to Y");  
}
```



→ Tip : define X et Y as memory labels in the .data segments

```
SECTION          .data  
x:               dw    5  
y:               dw    10
```

# Typical Construct: If

```
mov     ax, [x]           ; need to use ax as manipulating words
cmp     ax, [y]           ; (16-bit values)
jge     sinon             ; do something if x<y
...
jmp     continue
sinon:  ...                ; something else if x≥y
continue: ...              ; leaving if then else block
```

# Typical Construct: If

```
mov     ax, [x]           ; need to use ax as manipulating words
cmp     ax, [y]           ; (16-bit values)
jge     sinon
mov     rsi,msg1          ; select message 1 to be printed
mov     rdx,len1
jmp     continue
sinon:
mov     rsi,msg2          ; select message 2 to be printed
mov     rdx,len2
continue:
mov     rax, 1            ; system call for write
mov     rdi, 1            ; file handle 1 is stdout
syscall          ; invoke operating system to do the write
end:
...
```

# Typical Construct: If

```
SECTION .data
x:      dw 50
y:      dw 10
msg1:   db 'x smaller than y', 10
len1:   equ $-msg1
msg2:   db 'x greater or equal to y', 10
len2:   equ $-msg2
SECTION .text
GLOBAL _start
_start:
    mov     ax, [x]           ; need to use ax as manipulating words
    cmp     ax, [y]           ; (16-bit values)
    jge     sinon
    mov     rsi,msg1          ; select message 1 to be printed
    mov     rdx,len1
    jmp     continue
sinon:
    mov     rsi,msg2          ; select message 2 to be printed
    mov     rdx,len2
continue:
    mov     rax, 1            ; system call for write
    mov     rdi, 1            ; file handle 1 is stdout
    syscall                  ; invoke operating system to do the write
end:
...
```

# Addressing Modes

- Refers to how operands are obtained
- We have seen three so far
  - Immediate (`mov rax, 0x8`)
  - Register (`mov rax, rbx`)
  - Direct (`mov rax, [100]`)
- But there are more !

# Addressing Modes (cont.)

## ■ Not involving memory

→ Immediate (or literal) : `mov rax,0x41`

→ Register : `mov rax, rbx`

Reminder

Reminder

## ■ Involving memory

→ Direct : `mov rax,[0x16]` ; 0x16 can be a label

→ Indirect: look up address position in a register

• `mov rax,[rbx]` ; [rbx]->memory starting at rbx

→ Indirect with displacement

• `mov rax,[rbx+10]` ; also `rbx[10]`, `10[rbx]`

→ General case

• `mov rax,[rbx+scale*rsi+10]` ; `rbx[10+rsi*scale]`

scale ∈ {1,2,4,8}

Constant can be a label

Reminder



# Addressing Modes (cont.)

- If `message` and `len` contains the following  
`message: db 'HELLOWORLD', 10`  
`len: equ $-message`
- How would you downcase the whole string?  
→ Tip: use a loop on `rsi`, use indexed addressing



# Addressing Modes (cont.)

```
    mov     rsi,9
loop:
    add     BYTE [message+rsi], 'a' - 'A'
    dec     rsi
    jge     loop
```

- How would you do if the string contains a space?

→ `message:`      `db 'HELLO WORLD',10`



# Addressing Modes (cont.)

```
    mov     rsi,9
loop:
    cmp     BYTE [message+rsi], ' '
    je      cont
    add     BYTE [message+rsi], 'a' - 'A'
cont:
    dec     rsi
    jge     loop
```

- Homework: strings containing
  - any non-letters, and lowercase letters
  - E.g. `message: db 'HELLO World !!??',10`




# Summary

At the end of this session you should be able to


- be able to translate in assembly simple control flow constructs (loops, if-then-else statements)
- understand the main addressing modes of the x86
- draft simple programs in assembly code

# Further Reading



- <http://www.charlespetzold.com/code/>


# Even Further Reading



- <http://simonsingh.net/books/the-code-book/>

# Bonus Material

- Not exam material



# Operands of different sizes

- ~~mov~~ rax, BYTE [message]


- Does not work: “error: mismatch in operand sizes”

- Reason: does not know how to set extra bits of rax

- Two extra **mov** instructions

- **movzx** Sets extra bits to zero

- **movsx** Sets extra bits to bit of sign (for signed values)






# hello\_world.asm on DOS

```
.MODEL SMALL          ; memory model
.STACK                ; stack segment
.DATA                 ; data segment
    MESSAGE DB 'HELLO WORLD$'

.CODE
start:
    MOV AX, @DATA      ; initialising DS
    MOV DS, AX         ; (16bit only)

    MOV DX, offset MESSAGE ; DX points to string MESSAGE
    MOV AH, 09H        ; 09H = DOS print routine
    INT 21H            ; calling DOS print routine


    MOV AH, 4CH        ; 4CH = DOS exit routine
    INT 21H            ; calling DOS exit routine
END start
```



```
C:\> tasm.exe hello_world.asm
C:\> tlink.exe hello_world.obj
C:\> hello_world.exe
HELLO WORLD
C:\>
```

# Instruction INT 21h

- Interruptions
  - Causes code to jump to location stored in interrupt table
  - Triggered by HW, proc (/zero), or special op (INT)
  - Proc state saved on INT, restored on return (IRET)
- In 8086 / 8186 : INT 21h = "paleo-software"
  - Prog run in "real mode" = direct access to CPU
  - INT 21h : access to **DOS routines** (routine # in AH)
- $\geq$  8286 : privilege levels
  - Normal prog = run in **protected mode** (= user mode)
  - INT: intercepted by OS → can be used for syscalls
  - On x86-64 : SYSENTER & SYSCALL (faster)





<https://fr.wikipedia.org/wiki/Limulidae>